
1 Program Structure

prg ⇒ [ global pragma ]* [ ( module | class ) ] [ interface ]*
[ tfspec ]* [ structdef ]* [ typedef ]* [ objectdef ]*
[ function ]*

2 Module Declarations

module ⇒ module id [ deprecated str ] ;

class ⇒ class id [ deprecated str ] ; classtype

classtype ⇒ classtype type ;
| extern classtype ; [ interface pragma ]*

3 Import / Export

interface ⇒ ( import | use ) id : symbolset ;
| ( export | provide ) symbolset ;

symbolset ⇒ all [ except { ext id [ , ext id ] } ]
| { ext id [ , ext id ] }

4 Structure Definitions (future use)

structdef ⇒ struct id { [ type id [ , id ]* ; ]* } ;

5 Type Definitions

typedef ⇒ builtintypedef
| loctypedef
| exttypedef
| absttypedef
| typerel

builtintypedef ⇒ builtin typedef [ typecompspec ] id [ typeargs ] ;

loctypedef ⇒ typedef [ typecompspec ] type id ;

exttypedef ⇒ external typedef [ typecompspec ] id ; [ interface pragma ]*

absttypedef ⇒ typedef [ typecompspec ] id ;

typecompspec ⇒ [ id ]

typeargs ⇒ ( id [ , id ]* )

typerel ⇒ typerel id [ , id ]* <: id ;
| typerel id [ typeargs ] <: id [ typeargs ] [ | typerelexpr ] ;

typerelexpr ⇒ ( typerelexpr )
| typerelexpr <= typerelexpr
| typerelexpr >= typerelexpr
| typerelexpr == typerelexpr
| typerelexpr && typerelexpr
| id

6 Object Definitions

objectdef ⇒ ( locobjdef | extobjdef )

locobjdef ⇒ objdef type id = funcall ;

extobjdef ⇒ external objdef type id ; [ interface pragma ]*

7 Function Declarations and Definitions

function ⇒ extfundec [ ( interface pragma | funtion pragma ) ]*
| specfundec [ function pragma ]*
| fundef
| main

extfundec ⇒ external varsignature ;

specfundec ⇒ specialize fixsignature ;

fundef ⇒ [ inline ] [ thread ] fixsignature [ function pragma ]* body

fixsignature ⇒ fixrets ext id ( fixargs )
| operator sig

varsignature ⇒ varrets ext id ( varargs )
| operator sig

operator sig ⇒ type ( ext op ) ( arg )
| type ( ext op ) ( arg , arg )

fixargs ⇒ ( arg [ , arg ]* | [ void ] )

varargs ⇒ fixargs
| arg [ , arg ]* , ...

arg ⇒ type [ & ] id

fixrets ⇒ ( type [ , type ]* | [ void ] )

varrets ⇒ fixrets
| type [ , type ]* , ...

main ⇒ int main ( [ void ] ) body

3



8 Function Bodies

body ⇒ { [ cachesim pragma ] [ vardec ]* [ statement ]* [ return ] }
vardec ⇒ type id [ , id ]* ;

statement ⇒ ;
| assignment ;
| funcall ;
| withloop ;
| cond
| doloop
| whileloop
| forloop

return ⇒ return [ expr ] ;
| return ( [ exprs ] ) ;

assignment ⇒ assign lhs [ , assign lhs ]* assign op expr
| assign lhs ( ++ | -- )

assign lhs ⇒ id
| assign lhs . id
| assign lhs [ exprs ]

assign op ⇒ ( = | += | -= | *= | /= | %= )

cond ⇒ if ( expr ) statementblock [ else statementblock ]

doloop ⇒ do statementblock while ( expr ) ;

whileloop ⇒ while ( expr ) statementblock

forloop ⇒ for ( assignment [ , assignment ]*
; expr ; assignment [ , assignment ]* )

statementblock

statementblock ⇒ { [ cachesim pragma ] [ statement ]* }
| statement

9 Expressions

exprs ⇒ expr [ , expr ]*

expr or dot ⇒ ( expr | . )

expr or mdot ⇒ ( expr | . | ... )

expr ⇒ const
| qual ext id
| funcall
| withloop
| set notation
| array
| struct
| expr || expr
| expr && expr
| expr ? expr : expr
| ( type ) expr
| ( expr )

arrray ⇒ [ [ exprs ] ]
| [ : type ]
| expr [ [ expr or mdot [ , expr or mdot ]* ] ]

struct ⇒ < exprs >
| expr . id

funcall ⇒ qual ext id ( [ exprs ] )
| spawn [ ( str ) ] qual ext id ( [ exprs ] )
| rspawn [ ( str ) ] qual ext id ( [ exprs ] )
| unary prf ( expr )
| qual ext op expr
| binary prf ( expr , expr )
| expr qual ext op expr
| ternary prf ( expr , expr , expr )

set notation ⇒ { id -> expr }
| { [ [ id or mdot [ , id or mdot ]* ] ] -> expr }

10 With-Loops

withloop ⇒ [ local ] with [ generators ] : operations

generators ⇒ { [withloop pragma ] [ generator ]* }
generator ⇒ ( index set ) [ { [ statement ]* } ] : gen exprs ;

index set ⇒ expr or dot ( < | <= ) index vars ( < | <= ) expr or dot
[ step expr [ width expr ] ]

index vars ⇒ id [ = id vec ]
| id vec

id vec ⇒ [ [ id [ , id ]* ] ]

gen exprs ⇒ void
| expr
| ( expr [ , expr ]* )

operations ⇒ void
| operation
| ( operation [ , operation ]* )

operation ⇒ genarray ( expr [ , expr ] )
| modarray ( expr )
| fold ( ( qual ext id | qual ext op ) [ ( exprs ) ] , expr )
| foldfix ( ( qual ext id | qual ext op ) [ ( exprs ) ] , expr , expr )
| propagate ( id )

11 Types

type ⇒ basetype [ shape spec ]

shape spec ⇒ [ * ]
| [ + ]
| [ [ . [ , . ]* ] ]
| [ [ num [ , num ]* ] ]

basetype ⇒ simpletype
| usertype

simpletype ⇒ byte
| short
| int
| long
| longlong
| ubyte
| ushort
| uint
| ulong
| ulonglong
| float
| bool
| char
| double

usertype ⇒ ( [ struct ] | id :: ) id

polytype ⇒ < id [ = id [ id ] ] >
| < id ( -> | <- ) id [ id ] >

12 Identifiers

id or mdot ⇒ ( id | . | ... )

qual ext id ⇒ [ id :: ] ext id

ext id ⇒ ( id | reservedid )

reservedid ⇒ genarray
| modarray
| fold
| foldfix
| propagate
| all
| except

qual ext op ⇒ [ id :: ] ext op

ext op ⇒ ( op | reservedop )

reservedop ⇒ &
| &&
| ||
| !
| ~
| +
| -
| *
| /
| %
| <=
| <
| >=
| >
| >>
| <<
| ^
| ++
| --

13 Constants

const ⇒ numbyte
| numshort
| numint
| numlong
| numlonglong
| numubyte
| numushort
| numuint
| numulong
| numulonglong
| num
| float
| double
| char
| [ str ]+

| true
| false

14 Builtin Operations

unary prf ⇒ ( tob S | tos S | toi S | tol S | toll S )
| ( toub S | tous S | toui S | toul S | toull S )
| tof S
| tod S
| toc S
| tobool S
| ( not S | not V )
| ( neg S | neg V )
| ( abs S | abs V )
| dim A
| shape A

ternary prf ⇒ modarray AxVxS

binary prf ⇒ ( add SxS | add SxV | add VxS | add VxV )
| ( sub SxS | sub SxV | sub VxS | sub VxV )
| ( mul SxS | mul SxV | mul VxS | mul VxV )
| ( div SxS | div SxV | div VxS | div VxV )
| ( mod SxS | mod SxV | mod VxS | mod VxV )
| ( min SxS | min SxV | min VxS | min VxV )
| ( max SxS | max SxV | max VxS | max VxV )
| ( eq SxS | eq SxV | eq VxS | eq VxV )
| ( neq SxS | neq SxV | neq VxS | neq VxV )
| ( le SxS | le SxV | le VxS | le VxV )
| ( lt SxS | lt SxV | lt VxS | lt VxV )
| ( ge SxS | ge SxV | ge VxS | ge VxV )
| ( gt SxS | gt SxV | gt VxS | gt VxV )
| ( and SxS | and SxV | and VxS | and VxV )
| ( or SxS | or SxV | or VxS | or VxV )
| reshape VxA
| sel VxA
| take SxV
| drop SxV
| cat VxV

secret prf ⇒ hideValue SxA
| hideShape SxA
| hideDim SxA
| sel VxIA
| ( reciproc S | reciproc V )
| mask VxVxV
| ( non neg val S | non neg val V )
| ( val le val SxS | val le val VxV )

15 Pragmas

global pragma ⇒ # pragma main-aspect qual ext id [ , qual ext id ]*

interface pragma⇒ # pragma linkname str
| # pragma cudalinkname [ str ]+

| # pragma linkwith [ str ]+

| # pragma linkobj [ str ]+

| # pragma copyfun str
| # pragma freefun str
| # pragma linksign [ nums ]
| # pragma refcounting [ nums ]
| # pragma effect qual ext id [ , qual ext id ]*

withloop pragma⇒ # pragma wlcomp funcall

cachesim pragma⇒ # pragma cachesim [ str ]*

function pragma⇒ # pragma recountdots
| # pragma mutcthreadfun
| # pragma noinline

